



# 16 Maple™

---

## Übersicht

16.1	Differentialoperatoren mit Maple™	208
16.2	Potential und Arbeitsintegral mit Maple™	209
16.3	Vektorpotential und Flussintegral mit Maple™	210
16.4	Differentialgleichungen erster Ordnung mit Maple™	211
16.5	Schwingungsdifferentialgleichung mit Maple™	212
16.6	Richtungsfeld und numerische Lösung eines Differentialgleichungssystems mit Maple™	213
16.7	Fourier-Analyse mit Maple™	214
16.8	Residuen und Laurent-Entwicklung mit Maple™	215

---

## 16.1 Differentialoperatoren mit Maple™

Bilden Sie für die Vektorfelder

$$\vec{F} = (x^2, y^2, z^2)^t, \quad \vec{G} = (y, z, x)^t$$

$U := \vec{F} \cdot \vec{G}$ ,  $\vec{H} := \vec{F} \times \vec{G}$  und bestimmen Sie  $\text{grad}U$ ,  $\Delta U$ ,  $\text{div} \vec{H}$ ,  $\text{rot} \vec{H}$ .

**Verweise:** [Rechenregeln für Differentialoperatoren](#)

### Lösungsskizze

(i) Definition der Felder:

```
> with(VectorCalculus): # Einbinden relevanter Funktionen
> SetCoordinates('cartesian' [x,y,z]):
> F := VectorField(<x^2,y^2,z^2>): G := VectorField(<y,z,x>):
> U := DotProduct(F,G): H := CrossProduct(F,G):
```

$$U := x^2y + y^2z + xz^2$$

$$H := (xy^2 - z^3)\vec{e}_x + (-x^3 + yz^2)\vec{e}_y + (x^2z - y^3)\vec{e}_z$$

(ii) Gradient und Laplace-Operator:

```
> gradU := Gradient(U): DeltaU := Laplacian(U):
```

$$\text{grad}U := (2xy + z^2)\vec{e}_x + (x^2 + 2yz)\vec{e}_y + (2xz + y^2)\vec{e}_z$$

$$\Delta U := 2x + 2y + 2z$$

```
> # alternativ (Def. von Delta):
```

```
> DeltaU := Divergence(gradU):
```

(iii) Divergenz und Rotation:

```
> divH := Divergence(H): rotH := Curl(H):
```

$$\text{div}H := x^2 + y^2 + z^2$$

$$\text{rot}H := (-3y^2 - 2yz)\vec{e}_x + (-2xz - 3z^2)\vec{e}_y + (-3x^2 - 2xy)\vec{e}_z$$

```
> # alternativ (Produktregeln für Differentialoperatoren):
```

```
> rotF := Curl(F): rotG := Curl(G):
```

$$\text{rot}F := (0)\vec{e}_x + (0)\vec{e}_y + (0)\vec{e}_z$$

$$\text{rot}G := (-1)\vec{e}_x + (-1)\vec{e}_y + (-1)\vec{e}_z$$

```
> divH := DotProduct(G,rotF)-DotProduct(F,rotG):
```


```
> # -> gleiches Resultat
```

## 16.2 Potential und Arbeitsintegral mit Maple™

Untersuchen Sie, ob die Vektorfelder

$$\vec{F} = (xy^2, yx^2)^t, \quad \vec{G} = (x^2y, y^2x)^t$$

ein Potential besitzen, bestimmen Sie es gegebenenfalls und berechnen Sie für beide Vektorfelder jeweils das Arbeitsintegral für den abgebildeten Weg.



**Verweise:** [Existenz eines Potentials](#), [Arbeitsintegral](#)

### Lösungsskizze

(i)  $\vec{F} = (xy^2, yx^2)^t$ :

```
> with(VectorCalculus): # Einbinden relevanter Funktionen
> SetCoordinates('cartesian'[x,y]):
> F := VectorField(<x*y^2,y*x^2>):
> # Prüfen der Integrabilitätsbedingung
> diff(F[1],y)-diff(F[2],x);
```

$$0 \quad (\implies \exists \text{Potential } U, \quad \vec{F} = \text{grad } U)$$

```
> U := ScalarPotential(F);
```

$$U := \frac{x^2y^2}{2}$$

```
> # Berechnung des Arbeitsintegrals als Potentialdifferenz
> AF := subs(x=2,y=2,U)-subs(x=2,y=0,U); # U(2,2)-U(2,0)
AF := 8
```

(ii)  $\vec{G} = (x^2y, y^2x)^t$ :

```
> G := VectorField(<x^2*y,y^2*x>):
> # Prüfen der Integrabilitätsbedingung
> diff(G[1],y)-diff(G[2],x);
```

$$x^2 - y^2 \quad (\neq 0 \implies \exists \text{kein Potential})$$

```
> # Parametrisierung der drei Teilwege
> C1 := Line(<2,0>,<0,2>): # Geradensegment (2,0)->(0,2)
> # Kreismittelpunkt und Radius
> mx := -1: my := 1: r := sqrt(2):
> C2 := Path(<mx+r*cos(t),my+r*sin(t)>,t=Pi/4..7*Pi/4):
> C3 := Line(<0,0>,<2,2>):
> # Berechnung der entsprechenden drei Arbeitsintegrale
> A1:=LineInt(G,C1); A2:=LineInt(G,C2); A3:=LineInt(G,C3);
```

$$A1 := 0, \quad A2 := -2/3, \quad A3 := 8$$

↪ gesamtes Arbeitsintegral  $AG := A1 + A2 + A3 = 22/3$

### 16.3 Vektorpotential und Flussintegral mit Maple™

Untersuchen Sie, ob die Vektorfelder

$$\vec{F} = (y^3, z, x^2)^t, \quad \vec{G} = (x, y^2, z^3)^t$$

ein Vektorpotential besitzen, bestimmen Sie es gegebenenfalls und berechnen Sie für beide Vektorfelder jeweils den Fluss durch die Halbkugelschale  $S : x^2 + y^2 + z^2 = R^2, z \geq 0$ , nach oben.

**Verweise:** [Vektorpotential](#), [Flussintegral](#)

#### Lösungsskizze

(i)  $\vec{F} = (y^3, z, x^2)^t$ :

```
> with(VectorCalculus): # Einbinden relevanter Funktionen
> SetCoordinates('cartesian'[x,y,z]):
> F := VectorField(<y^3,z,x^2>):
> # Überprüfung der Divergenzfreiheit (notw. Bed.)
> Divergence(F);
```

$$0 \quad (\implies \exists \text{Vektorpotential } \vec{A}, \quad \vec{F} = \text{rot } \vec{A})$$

```
> A := VectorPotential(F);
```

$$A := (z^2/2 - x^2y)\vec{e}_x + (-y^3z)\vec{e}_y + (0)\vec{e}_z$$

Satz von Stokes  $\implies$

$$\iint_S \underbrace{\text{rot } \vec{A}}_{\vec{F}} \cdot d\vec{S} = \int_C \vec{A} \cdot d\vec{r}, \quad C : x^2 + y^2 = R^2, z = 0$$

(Fluss nach oben, Kreis entgegen dem Uhrzeigersinn orientiert)

```
> # Randkurve der Hemisphäre -> Kreis in der xy-Ebene
> C := Path(<R*cos(phi),R*sin(phi),0>,phi=0..2*Pi):
> # Berechnung des Flussintegrals als Arbeitsintegral
> Fluss := LineInt(A,C);
```

$$\text{Fluss} := \frac{\pi}{4} R^4$$

(ii)  $\vec{G} = (x, y^2, z^3)^t$ :

```
> G := VectorField(<x,y^2,z^3>):
```

```
> # Überprüfung der Divergenzfreiheit (notw. Bed.)
```

```
> Divergence(G);
```

$$1 + 2y + 3z^2 \quad (\neq 0 \implies \exists \text{kein Vektorpotential})$$

```
> # Parametrisierung der Hemisphäre
```

```
> S := Surface(<R*sin(theta)*cos(phi),R*sin(theta)*sin(phi),
R*cos(phi)>,theta=0..Pi/2,phi=0..2*Pi):
```

```
> Fluss := Flux(G,S);
```

$$\text{Fluss} := \frac{2}{5}\pi R^5 + \frac{2}{3}\pi R^3$$

## 16.4 Differentialgleichungen erster Ordnung mit Maple™

Bestimmen Sie die allgemeinen Lösungen der Differentialgleichungen

$$\text{a) } y' = \frac{y}{x^2 + 1} \qquad \text{b) } y' = \frac{x}{y^2 + 1}$$

sowie die Lösungen zu dem Anfangswert  $y(1) = 2$ .

Verweise: [Maple™-Funktionen](#)

### Lösungsskizze

```
> with(DEtools): # Einbinden relevanter Funktionen
```

```
a) y' = y/(x^2 + 1), y(1) = 2:
```

```
> DG := diff(y(x), x) = y(x)/(x^2+1):
```

```
> # allgemeine Lösung
```

```
> Lsg := dsolve(DG);
```

$$Lsg := y(x) = \_C1 e^{\arctan(x)}$$

```
> # Lösung des Anfangswertproblems
```

```
> IVPsol(y(1)=2,Lsg);
```

$$y(x) = \frac{2e^{\arctan(x)}}{e^{\pi/4}}$$

```
b) y' = x/(y^2 + 1), y(1) = 2:
```

```
> DG := diff(y(x), x) = x/(y(x)^2+1):
```

```
> # Typ der Differentialgleichung
```

```
> odeadvisor(DG)
```

[ \_separable ]

```
> # allgemeine Lösung (implizite Darstellung)
```

```
> dsolve(DG, 'implicit');
```

$$\frac{x^2}{2} - \frac{y(x)^3}{3} - y(x) + \_C1 = 0$$

```
> # Lösung des Anfangswertproblems
```

```
> dsolve({y(1)=2,DG}, 'implicit');
```

$$\frac{x^2}{2} - \frac{y(x)^3}{3} - y(x) + \frac{25}{6} = 0$$

Etwas schwieriger: Lösen Sie  $y' = 1 + x/y$  (ohne `dsolve`); eventuell mit der Hilfe der Funktion `odeadvisor`, die die Klassifizierung

```
[[_homogeneous, class A], _rational, [_Abel, 2nd type, class A]]
```

liefert.

## 16.5 Schwingungsdifferentialgleichung mit Maple™

Bestimmen Sie die allgemeine Lösung der Schwingungsdifferentialgleichung

$$u'' + 4u' + 3u = \cos(\omega t)$$

sowie, für  $\omega = 1$ , die Lösung  $u_p$  zu den Anfangswerten  $u(0) = 1$ ,  $u'(0) = 0$ .  
Zeichnen Sie  $u_p$  auf dem Intervall  $[0, 4\pi]$ .

Verweise: [Differentiation mit Maple™](#)


### Lösungsskizze

```
> with(DEtools): # Einbinden relevanter Funktionen
> DG := D^(2)(u)(t)+4*D(u)(t)+3*u(t) = cos(omega*t):
> # allgemeine Lösung
> allg_Lsg := dsolve(DG);
```

$$\text{allg\_Lsg} := u(t) = e^{-3t} \_C2 + e^{-t} \_C1 + \frac{1}{\omega^4 + 10\omega^2 + 9} \left( -\omega^2 \cos(\omega t) + 4\omega \sin(\omega t) + 3 \cos(\omega t) \right)$$

```
> # partikuläre Lösung und Grafik
> omega := 1:
> u_p := IVPsol([u(0)=1,D(u)(0)=0],allg_Lsg);
> DEplot(DG,u(t),t=0..4*Pi,[u(0)=1,D(u)(0)=0]);
```

$$u_p := u(t) = -\frac{7e^{-3t}}{20} + \frac{5e^{-t}}{4} + \frac{\cos(t)}{10} + \frac{\sin(t)}{5}$$



starke Dämpfung  $\rightarrow$

schnelles Abklingen des nichtperiodischen Anteils

## 16.6 Richtungsfeld und numerische Lösung eines Differentialgleichungssystems mit Maple™

Zeichnen Sie das Richtungsfeld des Differentialgleichungssystems


$$u' = \cos(u) + \sin(u + v), \quad v' = \sin(v) + \cos(u + v)$$

in  $[0, 2\pi]^2$  sowie die Lösungskurve  $(u_*(t), v_*(t))$ ,  $t \in [0, 10]$ , zu den Anfangsbedingungen  $u_*(0) = 4$ ,  $v_*(0) = 2$ . Bestimmen Sie numerisch die Kurvenpunkte für  $t = 5, 10$ .

Verweise: [Differentiation mit Maple™](#)

### Lösungsskizze

```
> with(DEtools): # Einbinden relevanter Funktionen
> # Differentialgleichungen und Anfangsbedingungen
> DG1 := D(u)(t) = cos(u(t))+sin(u(t)+v(t)):
  DG2 := D(v)(t) = sin(v(t))+cos(u(t)+v(t)):
> AB1 := u(0) = 4: AB2 := v(0) = 2:
> # Richtungsfeld mit Lösungskurve
> DEplot([DG1,DG2],[u(t),v(t)],t=0..10,u=0..2*Pi,v=0..2*Pi,
  [[AB1,AB2]]):
```



```
> # numerische Lösung für t = 5,10
> T := [5,10]:
> dsolve({DG1,DG2,AB1,AB2},numeric,output=Array(T));
```

$$\begin{bmatrix} \begin{bmatrix} t & u(t) & v(t) \end{bmatrix} \\ \begin{bmatrix} 5. & 0.7753 & 3.2729 \end{bmatrix} \\ \begin{bmatrix} 10. & 1.3520 & 1.9689 \end{bmatrix} \end{bmatrix}$$

## 16.7 Fourier-Analyse mit Maple™

Bestimmen Sie die Fourier-Transformierte der Funktion

$$f(x) = \cos(x) \exp(-|x|)$$

und illustrieren Sie die Gültigkeit des Satzes von Plancherel und der Poisson-Summutationsformel.

**Verweise:** [Fourier-Transformation](#), [Satz von Plancherel](#), [Poisson-Summutationsformel](#)

### Lösungsskizze

(i) Fourier-Transformierte  $F(y) = \int_{\mathbb{R}} f(x) e^{-ixy} dx$ :

```
> with(inttrans): # Einbinden relevanter Funktionen
> f := x -> cos(x)*exp(-abs(x));
  F := fourier(f(x),x,y);
```

$$f := x \mapsto \cos(x)e^{-|x|}, \quad F := \frac{2(y^2 + 2)}{(y^2 - 2y + 2)(y^2 + 2y + 2)}$$

(ii) Satz von Plancherel  $2\pi \int_{\mathbb{R}} |f(x)|^2 dx = \int_{\mathbb{R}} |F(y)|^2 dy$ :

```
> If := int(f(x)^2,x=-infinity..infinity);
  IF := int(F^2,y=-infinity..infinity);
```

$$If := \frac{3}{4}, \quad IF := \frac{3\pi}{2} \quad \checkmark$$

(iii) Poisson-Summutationsformel  $\sum_{k=-\infty}^{\infty} f(k) = \sum_{k=-\infty}^{\infty} F(2\pi k)$ :

```
> Sf := sum(f(k),k=-infinity..infinity);
  SF := sum(subs(y=2*Pi*k,F),k=-infinity..infinity);
```

$$Sf := \frac{e^2 - 1}{(e^{1+i} - 1)(e^{1-i} - 1)}, \quad SF := \frac{\sinh(1/2) \cosh(1/2)}{\cos(1/2)^2 - \cosh(1/2)^2}$$

$Sf - SF = 0!$  ✓

Sieht (sehr) schwierig aus - aber die Formeln von Euler-Moivre und für die geometrische Reihe machen es leicht:

$$2f(k) \stackrel{k \geq 0}{=} (e^{ik} + e^{-ik})e^{-k} = q_+^k + q_-^k, \quad q_{\pm} = e^{\pm i - 1}$$

und  $f(k) = f(-k)$ ,  $\sum_{k=0}^{\infty} q_{\pm}^k = \frac{1}{1 - q_{\pm}}$ .



## 16.8 Residuen und Laurent-Entwicklung mit Maple™

Bestimmen Sie die Residuen der Funktionen

$$f(z) = \frac{1 + z^4}{z + 2z^3 + z^5}, \quad g(z) = \frac{\exp(z/2) - \cos(2z)}{\sin(z)^2}$$

(an allen Polstellen) sowie jeweils die ersten drei Terme der Laurent-Entwicklungen in der Umgebung doppelter Polstellen.

Verweise: [Residuum](#), [Laurent-Reihe](#)

### Lösungsskizze

(i)  $f(z) = (1 + z^4)/(z + 2z^3 + z^5)$ :

```
> with(numapprox): # Einbinden relevanter Funktionen
> f := (1+z^4)/(z+2*z^3+z^5):
> # einfache Polstelle bei z=0
> R0 := residue(f,z=0);
```

$$R0 := 1$$

```
> # doppelte Polstellen bei z = -i, i
> LmI := laurent(f,z=-I,4); LpI := laurent(f,z=I,4);
```

$$LmI := -\frac{1}{2}(z+I)^{-2} + \frac{7I}{8} + O((z+I))$$

$$LpI := \frac{1}{2}(z-I)^{-2} - \frac{7I}{8} + O((z-I))$$

⇒ Residuen (Koeffizienten von  $(z \pm i)^{-1}$  der Laurent-Entwicklungen): für beide doppelten Polstellen null

(ii)  $g(z) = (\exp(z/2) - \cos(2z))/\sin(z)^2$ :

```
> g := z -> (exp(z/2)-cos(2*z))/sin(z)^2:
> # einfache Polstelle bei z=0
> R0 := residue(g(z),z=0);
```

$$R0 := 1/2$$

```
> # doppelte Polstellen bei z=k*Pi, k<>0
> assume(k,integer,k<>0)
> Lk := laurent(g(z),z=k*Pi,4):
> Lk := simplify(Lk); # Vereinfachung
```

$$Lk := \left( e^{k \sim \pi/2} - 1 \right) (z - k \sim \pi)^{-2} + \frac{1}{2} e^{k \sim \pi/2} (z - k \sim \pi)^{-1} + \frac{11e^{k \sim \pi/2}}{24} + \frac{5}{3} + O((z - k \sim \pi))$$

⇒ Residuen:  $\frac{1}{2}e^{k \sim \pi/2}$

“ $k \sim$ ” symbolisiert die Einschränkungen an die Variable  $k$  ( $k \in \mathbb{Z} \setminus \{0\}$ )